

# CURSOR

THE TINY MICRO COMPUTER NEWS SERVICE

VOLUME I

\$2.00 per Issue

MARCH 1980

\$9.50 Semi-Annually

ISSUE 3

## WORLD RECORD

We at CURSOR believe we have, through great perserverance, tenacity, prior planning, and a great deal of effort, captured the record for being late in publishing an issue!

We have grown weary of receiving the April issue of every periodical we subscribe to, in February. With this issue, we are starting a NEW trend, i.e., the March issue will be received in April! Ah, how "headv" success is!! (Could this per-

chance be a brilliant, albeit clumsy attempt to bilk our subscribers of 1 issue?). Fortunately, the truth is somewhat more mundane (excuse follows).

We received a program and assorted notes from subscriber Brett Bilbrey, which set us on fire and subsequently caused our issue to run several weeks late and moved our telephone bill into the realm of high finance (it is incomprehensible how several glitches can cause so much work). We are sure that your wait will have been worth it!

EDITORS NOTE: Address all "character assassinations", "snide comments", and vicious letters to:

Mr. Brett Bilbrey, 14430 Barclay, Dearborn MI 48126.

However, send all kind and praiseworthy comments to CURSOR (just kidding?).

Without further ado, we give you:

THREE VOICE  
MUSIC ASSEMBLER

BY

BRETT BILBREY

This program uses sound ports &(16) through &(19) and &(21) + &(22). Brett uses PEEK & POKE to transform REM statements into DATA lines (lines 10 through 22). Ed. NOTE: See Issue #2 for explanation of PEEK & POKE.

The program format allows you to write your own music, or if desired, merely input our examples. If you intend to input only our printed music, 2 REM statements (lines 10 and 11) will be required. Each REM line you

create is capable of storing a maximum of 24 chords or rests. The program cannot accept more than 13 REM lines (312 chords or rests).

If you attempt to input 35 chords and you are only using one REM statement, your computer will load the first 24 chords into that REM statement and then start altering and erasing the program body (in other words, it won't work).

Each REM statement contains a Line Number, a Period, and 97 characters (any characters will do). The easiest way to do this is to keep typing in "1234567890" over and over until your computer will not accept

any more characters (Line Input Buffer is filled). Then erase backwards, leaving the "7" as last line entry.

After you have input REM lines and program, you are ready to input the music. Referring to the MUSIC CHART for "Star Wars", press "RUN" and "GO"; the screen will clear and print "CHORD #1", space a line and print "A"; press "121", hit "GO", and "B" will appear; input "Ø", hit "GO", and "C" will appear; press "Ø", hit "GO", and "DURATION" will appear; press "16" and "GO", the screen



MAIL ADDRESS: P.O. BOX 266, NORTH HOLLYWOOD, CA 91603

BUSINESS ADDRESS: 6115 CLYBOURN, SUITE 25, NORTH HOLLYWOOD, CA 91606 • (213) 763-7701

will clear and print "CHORD #2", and "A", etc.

When song input has been completed, press "999", hit "GO" - the computer will save the end of song location in variable "E" and immediately start playing the music you have input (Jumps to Line 200).

Whenever you want to PLAY music, DO NOT HIT "RUN", instead, type "GOTO 200".

To save your songs on tape, it will be necessary to save variable "E" so your computer will end the song at the proper place; this can be done thusly:

```
KEY IN: NT=1;:PRINT ;TV=13;LIST ;PRINT #1,
        ":RETURN ;E=",E;PRINT "GOTO 200"
```

DO NOT HIT GO! Start tape recorder in "RECORD" mode and then hit "GO". Your program, music, and variable "E" will be stored on tape.

To input this tape merely press ":INPUT", hit "GO", press PLAY on tape recorder. Program and music will load and automatically play music.

### THREE VOICE MUSIC ASSEMBLER

BY

BRETT BILBREY

```
10 .12345678901234567890123456789012345678
901234567890123456789012345678901234567
89012345678901234567
11 .12345678901234567890123456789012345678
901234567890123456789012345678901234567
89012345678901234567
12 through 22 (exactly the same, as needed)
100 D=-24573
105 Y=1
110 FOR X=0 TO 92STEP 4
115 CLEAR
117 PRINT #1,"CHORD #",Y
120 FOR C=0 TO 3
130 IF C=0 INPUT "A"J;GOTO 170
140 IF C=1 INPUT "B"J;GOTO 170
150 IF C=2 INPUT "C"J;GOTO 170
160 INPUT "DURATION"J
170 IF J=999E=(D+X+C)-5;GOTO 200
175 J=J-127
180 IF J<0%(D+X+C)=J-1;GOTO 195
190 %(D+X+C)=J
195 NEXT C;Y=Y+1;NEXT X
197 D=D+101;GOTO 110
200 NT=0;&(16)=49;&(21)=15;&(22)=136;A=-24574
205 T=2
210 FOR C=0 TO A=92STEP 4
220 &(17)=%(C)÷256+127;&(18)=%(C+1)÷256+127;
&(19)=%(C+2)÷256+127
230 FOR D=1 TO %(C+3)÷256+127)xT;NEXT D
```

```
240 IF C>=EGOTO 250
245 NEXT C;A=A+101;GOTO 210
250 &(21)=0;&(22)=0;&(16)=0
260 &(17)=0;&(18)=0;&(19)=0;NT=3
```

### NOTE VALUE TABLES

NOTE DURATION (DU):

```
Whole note = 200
1/2 note = 100
1/4 note = 50
1/8 note = 25
Triplet = 16
1/16 note = 12
1/32 note = 6
```

#### LOW OCTIVE

BALLY NOTE	NOTE NAME	INPUT TONE #
÷2	D	244
	D Sh	230
÷3	E	215
÷4	F	204
	F Sh	192
÷5	G	181
	G Sh	170
÷6	A	160
	B Fl	151
÷7	B	143

#### MIDDLE OCTIVE

BALLY NOTE	NOTE NAME	INPUT TONE #
1	C	136
	C Sh	129
2	D	121
	D Sh	114
3	E	107
4	F	101
	F Sh	95
5	G	90
	G Sh	85
6	A	80
	A Sh	76
7	B	71

#### HIGH OCTIVE

BALLY NOTE	NOTE NAME	INPUT TONE #
x1	C	67
	C Sh	64
x2	D	60
	E Fl	57
x3	E	53
x4	F	50
	F Sh	47
x5	G	44
	G Sh	42

# HIGH OCTIVE (CONT)

BALLY NOTE	NOTE NAME	INPUT TONE #
x6	A	39
	B F1	37
x7	B	35
		33

# STAR WARS MUSIC BY MARION NELEPA

CHD	1	2	3	4	5	CHD
A	121	∅	121	143	9∅	A
B	∅	121	∅	121	71	B
C	∅	∅	∅	9∅	6∅	C
DU	16	16	16	1∅∅	1∅∅	DU
CHD	6	7	8	9	10	CHD
A	1∅7	1∅7	1∅7	71	9∅	A
B	9∅	9∅	9∅	6∅	71	B
C	67	71	8∅	44	6∅	C
DU	16	16	16	1∅∅	5∅	DU
CHD	11	12	13	14	15	CHD
A	1∅7	1∅7	1∅7	44	9∅	A
B	9∅	9∅	9∅	71	71	B
C	67	71	8∅	6∅	6∅	C
DU	16	16	16	1∅∅	5∅	DU
CHD	16	17	18	19	20	CHD
A	1∅1	1∅1	1∅1	121	999	A
B	8∅	8∅	8∅	95		B
C	67	71	67	8∅		C
DU	16	16	16	2∅∅		DU

NOTE: If you wish to change the speed at which the song is played, give "T" in Line 205 a higher value to slow down the speed, and a lower value to speed the song up.

For those that desire an additional music example, we have very rapidly thrown together a version of "Chopsticks". If you can't write a better version yourself, you must have a tin ear.

# CHOPSTICKS BY

FRED CORNETT

This program as written will play the refrain once, to play the refrain twice: CHANGE LINES 205 & 240 to read

205 T=2;F=∅

240 IF C>=EGOTO 247

ADD LINE 247:

247 F=F+1;IF F=1A=-24574;GOTO 210

When ready to input other music, change lines back to normal(as in listing).

CHD	1	2	3	4	5	CHD	
A	5∅	∅	5∅	∅	5∅	A	
B	44	∅	44	∅	44	B	
C	∅	∅	∅	∅	∅	C	
DU	5∅	∅	5∅	∅	5∅	DU	
CHD	6	7	8	9	10	CHD	
A	∅	5∅	∅	5∅	∅	A	
B	∅	44	∅	44	∅	B	
C	∅	∅	∅	∅	∅	C	
DU	∅	5∅	∅	5∅	∅	DU	
CHD	11	12	13	14	15	CHD	
A	5∅	∅	53	∅	53	A	
B	44	∅	44	∅	44	B	
C	∅	∅	∅	∅	∅	C	
DU	5∅	∅	5∅	∅	5∅	DU	
CHD	16	17	18	19	20	CHD	
A	∅	53	∅	53	∅	A	
B	∅	44	∅	44	∅	B	
C	∅	∅	∅	∅	∅	C	
DU	∅	5∅	∅	5∅	∅	DU	
CHD	21	22	23	24	25	CHD	
A	53	∅	53	∅	6∅	A	
B	44	∅	44	∅	35	B	
C	∅	∅	∅	∅	∅	C	
DU	5∅	∅	25	∅	5∅	DU	
CHD	26	27	28	29	30	CHD	
A	∅	6∅	∅	6∅	∅	A	
B	∅	35	∅	35	∅	B	
C	∅	∅	∅	∅	∅	C	
DU	∅	5∅	∅	5∅	∅	DU	
CHD	31	32	33	34	35	CHD	
A	6∅	∅	6∅	∅	6∅	A	
B	35	∅	35	∅	35	B	
C	∅	∅	∅	∅	∅	C	
DU	5∅	∅	5∅	∅	5∅	DU	
CHD	36	37	38	39	40	CHD	
A	∅	67	∅	67	∅	A	
B	∅	33	∅	33	∅	B	
C	∅	∅	∅	∅	∅	C	
DU	∅	2∅∅	∅	5∅	∅	DU	
CHD	41	42	43	44	45	CHD	
A	67	∅	6∅	∅	53	A	
B	33	∅	35	∅	39	B	
C	∅	∅	∅	∅	∅	C	
DU	1∅∅	∅	5∅	∅	53	DU	
CHD	46	999					CHD
A	∅						
B	∅						
C	∅						
D	∅						

## FIRST ANNUAL CURSOR MUSIC CONTEST

Now that you have played the music examples we have printed, and hopefully attempted to convert some of your favorite music, you have found that while converting and inputting music is somewhat time consuming, playing it back is very simple, and quite rewarding. We hereby open the "First Annual Cursor Music Contest". We will award a total of five prizes in four categories, with a "GRAND WINNER". The categories are: 1-CLASSICAL; 2-COUNTRY and WESTERN; 3-POPULAR; 4-ROCK. The four major category winners will receive a CURSOR T-Shirt, the GRAND WINNER will win a six month subscription to CURSOR and a CURSOR T-Shirt.

All ENTRIES MUST BE SUBMITTED ON CASSETTE TAPE, and postmarked no later than 5 May 80. Ties will be decided by complexity of music. All winning entries (and any others that are interesting) will be published.

When recording a program on tape, please use NT=1, a zero may work well on your machine, but causes input problems on other units.

If while putting your music entry on tape, you decide to submit a program or two (non-music), we will return your tape to you with a few of our own programs. Sounds like a fair deal doesn't it? (please allow approximately 3 weeks for return tape.)

\*\*\*\*\*

### GROWING PAINS

BY

FRED CORNETT  
MANAGING EDITOR

When CURSOR began, our intent was to create order out of havoc. We have, I believe, accomplished a great deal in a very short period of time; as witnessed by comparing the format and contents of this issue with Issue #1!

However, we find ourselves guilty of creating a certain amount of frustration and irritation among our readers. That irritation is caused by the time elapsed from the moment a subscriber mails CURSOR an order, to the time he receives the product. Please consider the following:

1. CURSOR is printed, not Xeroxed. Therefore, when we submit an issue or other printed matter to our printer, we are forced to "guesstimate" the amount of demand for this item. If we print too many, we lose money; If we print too few, we are forced to wait until enough orders are amassed to allow the item to be printed; thereby irritating the individual who has ordered the item.

With the printing of this issue, we believe we have finally reached a method that will please everyone. From this point onward, virtually all printed matter will be mailed within 14 days of order receipt.

2. DEADLINES: Contrary to popular belief, we DO set Issue deadlines. Our staff is small, with most of our time spent in administrative duties; research, advertising, filing, routine office procedure, etc. If each article, tutorial, and program printed in our monthly issues must be conceived, written, and formatted by CURSOR, the time requirement would be horrendous. We require your participation, your programs and your information to be able to adhere to a reasonable time frame. Go to your local newsstand and purchase a copy of "BYTE", "Kilobaud Microcomputing", or "Creative Computing" leaf through the pages, and see if you can find just one (1) program written by the staff of that magazine! Our staff members possess great creativity, but if their duties require the creation of programs for each issue, they won't have time for anything else. It is human nature to "let George do it"! Please don't wait for the other guy to send us his programs or articles, send us your program or article now, we need them!

We could easily meet our deadlines if we were willing to fill our pages with mediocre undocumented, ill-conceived information and programs. We understand your disappointment as the mailman once again fails to bring your copy of CURSOR. Believe me, we are working hard at eliminating this problem, but unfortunately, we only learn through experience. Your understanding and support will help.

This issue initiates a policy which we hope will motivate you to share your programs or ideas with the rest of our readers. For each program or article that you submit, which in turn is printed in CURSOR, you will receive either a subscription extension, or CURSOR products in exchange. This offer is extended only to those readers who do not advertise products within our pages (our only fee for advertisers is a shared program or tutorial). If the program you submit is of major order, i.e., a board game, major graphics package, applications program, or a major tutorial, the magnitude of products we give you will be sufficiently greater.

NOTE: As much as we would like to, it is impossible to personally answer all letters written to us. All programs and procedures mentioned in our advertisements have been printed, or will be printed within our pages,

including answers to your inquiries. Please keep the letters coming, we need your feedback and ideas. For those of you that need immediate answers, please give us a call, we will be happy to help you over the phone.

## CHICAGO LOOP

BY  
MIKE PEACE

Our thanks go to Mike Peace for sharing this simple but very innovative program. This program incorporates the use of three loops to provide a unique display of graphics looking very much like a city on a lake, complete with reflections, traffic and sound effects (LOVE IT!).

### PROGRAM EXPLANATION

110 Color  
120-160 Traffic  
190-230 Sets Sound Ports

### PROGRAM LISTING

```

1 .CHICAGO LOOP
2 .MIKE PEACE
10 CLEAR
20 FOR A=-80 TO 80
30 BOX A,0,5,RND (88),1
40 NEXT A
50 FOR A=1 TO 120
60 B=RND (160)-80
70 C=RND (40)
80 BOX B,C,1,1,2
90 BOX B,C-50,5,1,3
100 NEXT A
110 BC=14
120 FOR A=-80 TO 80
130 BOX A,0,2,1,RND (3)
140 BOX A+A,0,1,1,RND (3)
150 BOX 0-A,-3,2,1,RND (3)
160 BOX 0-A-A,-3,1,1,RND (3)
170 B=RND (20)
180 C=RND (20)
190 IF B<14&(21)=0
200 IF C<14&(22)=0
210 IF B>15&(21)=150
220 IF C>15&(22)=150
230 &(18)=42;&(19)=36
240 NEXT A
250 GOTO 120

```

## SCREEN FORMAT OUTPUT PORTS

If you wish to put more than 2 colors on-screen simultaneously (while using Tiny Basic) without resorting to machine language, it can only be done by formatting the screen. Port "&(9)" is the "Horizontal Blanking Register". By setting this port to var-

ious values, it will do a number of things  
1. Vertically split the screen; 2. Put up screen borders. Input this program example  
10 FOR A=0 TO 256;&(9)=A;PRINT A;NEXT A

The Color Ports are as follows:

Ports &(0) through &(3) control the colors used on the Right side of the screen (in conjunction with the use of &(9)).

Ports &(4) through &(7) control the colors used on the Left side of the Screen (in conjunction with the use of &(9)).

To use these ports, you assign them values just as you assign values to FC or BC.

Look at Lines 5 and 6 of the following program, and you will see how to assign value to these ports.

NOTE: "&(10)" is the "Vertical Blanking Register". Substitute &(10) for &(9) in the above one line program to see how it works

## LACE CURTAIN

```

5 BC=10;FC=83
6 &(2)=0;&(1)=53;&(0)=53
10 CLEAR ;&(9)=148;FOR A=-79 TO 79 STEP 2;BOX
A,0,1,87,1;NEXT A
20 FOR A=43 TO -43 STEP -2;BOX 0,A,159,1,1;NE
XT A;FOR A=1 TO 100;BOX 0,0,RND (159),RND
(87),3;NEXT A;RUN

```


The preceding program puts up 4 colors at the same time. By changing the value of &(, you will get 5 colors. Try experimenting!

## VIDEO BRAIN

Another system bites the dust!!! About one year ago, I very seriously contemplated the purchase of a Video Brain MicroComputer (F8 Microprocessor, uses APL/S as language) However, I thought the unit was over-priced at \$500.00 for the basic unit alone. Recently, our Los Angeles Bally Users Group found out that Video Brain went Bankrupt, and we were able to get a fantastic price on the basic units and cartridges by buying out the complete stock of a Texas dealer. To make a long story short, we have 3 units left, along with a full selection of cartridges, to include Checkers, Vice Versa, Music Teacher, Financier, Wordwise #1 & 2, Video Artist, Blackjack, Lemonade Stand, Pinball

We are offering those interested the complete unit; RF Adapter, AC Adapter, 2 Joysticks, + Gladiator Game Cartridge (with 384 games) for \$125.00 which includes shipping and tax Call or write for additional information.

Brand New in carton! Fred Cornett



TECHNICAL MANUALS  
AVAILABLE NOW

1. Bally On-Board ROM Sub-Routines. Which, in addition to the subroutines, include ASCII Standard & Nonstandard character sets, Cassette Memory Structure, Output Ports, Input Ports, Bally Data Base Locations, Bally Memory Locations, and On-Board ROM 8K Hex Dump. \$3.50 (+ 25c for 1st Class Postage)  
NOTE: All printed matter sent 3rd Class postage unless you designate otherwise.
2. Hackers Manual. Describes features provided in the Tiny Basic but not documented in the Bally Instruction Booklet. \$2.95
3. Disassembled Tiny Basic (CDOS Z80 Assembler version 02.15) \$6.50(+45¢ 1st Cl.)
4. Disassembled Brickyard & Clowns. \$6.95 (+ 45c for 1st Class Postage)
5. Disassembled DEMO Cassette \$6.50 (+ 45c for 1st Class Postage)
6. BALLY SYSTEM DESCRIPTION BOOK - Extensive and includes "Electrical Specification for Midway Custom Circuits", Timing, Interrupt handling explanations, etc. \$6.95 (+45¢ 1 C)
7. BALLY Service Manual - \$2.75 Schematics, etc.

All the above listed items are currently in stock and can be shipped immediately. (Money Order gets speediest response).

CHARACTER SET  
SIZE MULTIPLIER

BY  
FRED CORNETT

This program used POKE & CALL to generate character sets using multiplication factors of 2x, 4x or 8x. This program POKES a small machine language program (converted to decimal) to call up On-Board Sub Routine #52 (String Display Routine). See Feb 80 Issue for explanation of format.

After inputting the program, and pressing "RUN & GO", the computer will print "LETTER

SIZE ?" and then "INPUT 2,4 or 8". The program is asking you what size letters you wish to use; 2, 4 or 8 times normal size. After you input your selection, press "GO"; the screen will clear and wait for you to input characters. Try the following, input "HI!". After you have input the last character, press "GO"; screen will clear, and display letters in NEW size. When ready to start over, merely press any key.

NOTE: Be careful when using the 8x size, If you put in more characters than there is room on the screen for, the program will bomb!!!

```

10 CLEAR
15 PRINT "LETTER SIZE ?";INPUT "INPUT 2,4
   OR 8"L
20 IF L=2L=27672;GOTO 50
25 IF L=4L=-26600;GOTO 50
30 IF L=8L=-10216;GOTO 50
40 GOTO 15
50 CLEAR ;M=20180;N=M;G=125
65 P=-43;GOSUB G
70 P=53;GOSUB G
75 P=L;GOSUB G
80 P=20190;GOSUB G
85 P=-13871;GOSUB G
90 M=20190
95 C=KP;TV=C
100 IF C=13GOTO 110
105 %(M)=C;M=M+1;GOTO 95
110 %(M)=0
115 CLEAR
120 CALL (N);C=KP;GOTO 10
125 %(M)=P;M=M+2;RETURN

```

ROTATION

BY

ROBERT LEAKE

Editors Note: This program utilizes the "Plastic Puzzle"(CURSOR Feb 80) concept and moves it a few light years in complexity.

A group of 4 letters in a 2x2 square is rotated one position clockwise by keying in the letter in the upper left corner of the square. The object is to put a randomly selected board in alphabetic order (ABCD on top row, etc.) in as few moves as possible. In addition you have one special move which will interchange a pair of horizontally adjacent letters. To make a special move, key in "S", then the left letter of the pair. Specifying a letter in the right hand column will nullify the entry of the previous "S". If you make a mistake, think of a better strategy, or would otherwise like to start over, an "R" entry will restore the

original board. With each reset, you get another special move.

### PROGRAM EXPLANATION

- 10 Skips over subroutines
- 20- 30 Calculates CX,CY for board position P, prints letter
- 40- 50 Search-finds board pos. P for letter keyed in.
- 60 Resets board
- 70- 80 Time-delay, clear lower screen
- 90 Initialization, construct board.
- 100-120 Selects letters from A to P randomly (once each), puts them on board, stores board setup for use with reset.
- 130-140 Keypad input-Checks if correct.
- 150 Sees if input was "Q" (quit).
- 160 If input was "R" resets board, etc.
- 170 If input was "S" goto Special Moves
- 180 Search board for Input (A-P) goto Special Move (all regular moves are filtered through Special Move).
- 190-200 Insult anyone quitting & # of moves.
- 210-220 Start game over with/without same setup
- 230-270 Validation & execute Special moves.
- 280-290 " " " Regular " "
- 300-310 Sees is board is in correct order; if not-ask for another input.
- 310-320 Music & congratulations for winner.

### MAJOR VARIABLES

- L Counter-Special moves
- M ASCII corresponding to input
- N Counter-Total Moves
- P Board position (1-16; numbered from top)
- R Counter-Number of Resets
- S Flag-indicates previous input was "S"
- X Used in conjunction with reset, so that "MOVE # X?" is asking for move of initial game or move X of previous reset.

### PROGRAM LISTING

```
1 .ROTATION
10 GOTO 90
20 T=P+4;H=RM;IF H=0H=4
30 CX=-24+10xH;CY=45-10x((P-1)÷4+1);TV=@(P)
;RETURN
40 FOR A=1TO 16;IF M=@(A)P=A
50 NEXT A;RETURN
60 L=0;S=0;V=N;FOR B=33TO 48;P=B-32;@(P)=@(B);GOSUB 20;NEXT B;GOSUB 80;RETURN
70 FOR T=0TO 1500;NEXT T
80 CX=-27;CY=-20;BOX 0,-22,159,43,2;RETURN
90 L=0;N=0;R=0;S=0;V=0;CLEAR;FC=40;BC=223;FOR C=0TO 40STEP 10;BOX 0,C,41,1,1;BOX C-20,20,1,41,1;NEXT C
100 FOR Z=17TO 32;@(Z)=0;NEXT Z;FOR P=1TO 16
110 T=RND(16)+16;IF @(T)GOTO 110
120 @(P)=T+48;@(T)=1;@(P+32)=@(P);GOSUB 20;
```

NEXT P

```
130 CX=-27;CY=-30;PRINT "MOVE # ",#1,N-V+1
"?",;M=KP;IF M>63IF M<84TV=M;GOTO 150
140 GOTO 140
150 IF M=81GOTO 190
160 IF M=82R=R+1;GOSUB 60;CX=-27;CY=-10;PRINT "RESET # ",#1,R,"";GOTO 130
170 IF M=83GOTO 230
180 GOSUB 40;H=P-P:4x4=0;GOTO 230
190 GOSUB 80;FOR J=1TO 25;MU=59;NEXT J;PRINT "SPOILSPORT!"
200 GOSUB 70;CX=-66;PRINT "YOU QUIT AFTER",#1,N," MOVES!"
210 N=0;R=0;GOSUB 70;CX=-42;PRINT "NEW GAME?(Y/N)";IF KP#78RUN
220 GOSUB 60;GOTO 130
230 IF M=83S=1;GOTO 130
240 IF LGOTO 280
250 IF S=0GOTO 280
260 IF H S=0;GOTO 130
270 T=@(P);@(P)=@(P+1);@(P+1)=T;GOSUB 20;P=P+1;GOSUB 20;L=1;N=N+1;S=0;CX=-27;CY=-20;PRINT "SP. MOVE";GOTO 300
280 IF (P>11)+HGOTO 130
290 N=N+1;T=@(P);@(P)=@(P+4);@(P+4)=@(P+5);@(P+5)=@(P+1);@(P+1)=T;GOSUB 20;P=P+1;GOSUB 20;P=P+3;GOSUB 20;P=P+1;GOSUB 20
300 FOR E=1TO 15;IF @(E)>@(E+1)GOTO 130
310 NEXT E;GOSUB 80;NT=8;FOR A=1TO 2;MU=49;MU=51;MU=53;MU=98;MU=49;MU=48;MU=53;MU=98;MU=49;MU=48;MU=48;MU=48;MU=48;NEXT A
320 NT=3;CX=-60;PRINT "YOU WON IN",#1,N," MOVES!";GOTO 210
```



GOOD NEWS!!! Jack Nieman, formally the National Sales Manager for the Consumer Products Division of Bally Manufacturing Corporation, is in the process of setting up a National Distributor for the Arcade and associated products. Jack promised to provide complete hardware support for those of you that do not have a retail source for Bally products in your area. Cursor subscribers will receive a minimum discount of TEN PERCENT (10%) when you order from his firm.

S & W DISTRIBUTING  
5300 B McDERMOTT DR.  
BERKELEY, ILL. 60163  
(312)449-5000

We need all the support we can get!

CLASSIFIEDS & NOTICES


LOS ANGELES AREA BALLY USERS GROUP  
 TIME: Wednesday, 16 April 1980 7:30 PM  
 PLACE: 5640 Fair Avenue, Apt. 21  
 North Hollywood, CA 91601  
 HOST: Mr. Gary D. Caton  
 PHONE: (213) 763-0734

Jan-Feb  
 Back Issue  
 available.  
 \$1.60  
 EACH

There are six programs in this package, and 12 pages of totally NEW material! The last program is a 3-Dimensional Simulation worth the \$9.95 +75¢ postage & handling alone. This program simulates the feeling of being in an airplane, flying above the earth as the surface-plane moves towards you, the mountains in the distance get larger as you move toward them. As you move the joystick side to side, the horizon and mountains move in the opposite direction, keeping in full proportion to one another. Package contains Cartesian to XY conversions, video art, charts, listings of all programs, 3-D Simulations, and much much more!!!

Send \$9.95 + 75¢ postage & handling by check or Money Order to: Sebree's Computing, 456 Granite, Dept 3B, Monrovia, CA. 91016

SALE: 2 Games with excellent graphics. 1-Horse Race: 4 graphic horses, 5 races. #2-MAX: ROBOT will destroy you if you don't get him first! Author wrote CHICAGO LOOP




\$6.95 Box 2-301  
 FRANKSBURG, ILL. 60143

SEBREE'S COMPUTING

NOTE: The following advertised software is wholeheartedly recommended by CURSOR for any individual interested in generating complex graphics programs !!!

All of the photos on this page were taken using the graphic generator program available in this package. There is a special command (XY) that few are aware of. The intent of this software tutorial is to supply you with a working knowledge gained through this program and the very thorough documentation provided. The "XY" command requires much less memory than the standard way of positioning lines across the screen. Programs using this command properly will run twice as fast as those not using it.



**CURSOR**  
 P.O. BOX 266, NO. HOLLYWOOD, CA 91603

\*\*\*FIRST CLASS\*\*\*

\*\*FIRST CLASS\*\*